
NAG Library Routine Document

F08XAF (DGGES)

Note: before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold italicised terms and
other implementation-dependent details.

1 Purpose

F08XAF (DGGES) computes the generalized eigenvalues, the generalized real Schur form S; Tð Þ and,
optionally, the left and/or right generalized Schur vectors for a pair of n by n real nonsymmetric matrices
A;Bð Þ.

2 Specification

SUBROUTINE F08XAF (JOBVSL, JOBVSR, SORT, SELCTG, N, A, LDA, B, LDB, SDIM,
ALPHAR, ALPHAI, BETA, VSL, LDVSL, VSR, LDVSR, WORK,
LWORK, BWORK, INFO)

&
&

INTEGER N, LDA, LDB, SDIM, LDVSL, LDVSR, LWORK, INFO

REAL (KIND=nag_wp) A(LDA,*), B(LDB,*), ALPHAR(N), ALPHAI(N), BETA(N),
VSL(LDVSL,*), VSR(LDVSR,*), WORK(max(1,LWORK))

&

LOGICAL SELCTG, BWORK(*)

CHARACTER(1) JOBVSL, JOBVSR, SORT

EXTERNAL SELCTG

The routine may be called by its LAPACK name dgges.

3 Description

The generalized Schur factorization for a pair of real matrices A;Bð Þ is given by

A ¼ QSZT, B ¼ QTZT,

where Q and Z are orthogonal, T is upper triangular and S is upper quasi-triangular with 1 by 1 and 2 by
2 diagonal blocks. The generalized eigenvalues, �, of A;Bð Þ are computed from the diagonals of S and T
and satisfy

Az ¼ �Bz,
where z is the corresponding generalized eigenvector. � is actually returned as the pair �; �ð Þ such that

� ¼ �=�
since �, or even both � and � can be zero. The columns of Q and Z are the left and right generalized
Schur vectors of A;Bð Þ.
Optionally, F08XAF (DGGES) can order the generalized eigenvalues on the diagonals of S; Tð Þ so that
selected eigenvalues are at the top left. The leading columns of Q and Z then form an orthonormal basis
for the corresponding eigenspaces, the deflating subspaces.

F08XAF (DGGES) computes T to have non-negative diagonal elements, and the 2 by 2 blocks of S
correspond to complex conjugate pairs of generalized eigenvalues. The generalized Schur factorization,
before reordering, is computed by the QZ algorithm.

F08 – Least-squares and Eigenvalue Problems (LAPACK) F08XAF

Mark 24 F08XAF.1



4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A,
Hammarling S, McKenney A and Sorensen D (1999) LAPACK Users’ Guide (3rd Edition) SIAM,
Philadelphia http://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) Matrix Computations (3rd Edition) Johns Hopkins University Press,
Baltimore

5 Parameters

1: JOBVSL – CHARACTER(1) Input

On entry: if JOBVSL ¼ N , do not compute the left Schur vectors.

If JOBVSL ¼ V , compute the left Schur vectors.

Constraint: JOBVSL ¼ N or V .

2: JOBVSR – CHARACTER(1) Input

On entry: if JOBVSR ¼ N , do not compute the right Schur vectors.

If JOBVSR ¼ V , compute the right Schur vectors.

Constraint: JOBVSR ¼ N or V .

3: SORT – CHARACTER(1) Input

On entry: specifies whether or not to order the eigenvalues on the diagonal of the generalized Schur
form.

SORT ¼ N
Eigenvalues are not ordered.

SORT ¼ S
Eigenvalues are ordered (see SELCTG).

Constraint: SORT ¼ N or S .

4: SELCTG – LOGICAL FUNCTION, supplied by the user. External Procedure

If SORT ¼ S , SELCTG is used to select generalized eigenvalues to the top left of the generalized
Schur form.

If SORT ¼ N , SELCTG is not referenced by F08XAF (DGGES), and may be called with the
dummy function F08XAZ.

The specification of SELCTG is:

FUNCTION SELCTG (AR, AI, B)

LOGICAL SELCTG

REAL (KIND=nag_wp) AR, AI, B

1: AR – REAL (KIND=nag_wp) Input
2: AI – REAL (KIND=nag_wp) Input
3: B – REAL (KIND=nag_wp) Input

On entry: an eigenvalue ARðjÞ þ
ffiffiffiffiffiffiffi
�1
p

� AIðjÞ
� �

=BðjÞ is selected if
SELCTG ARðjÞ;AIðjÞ;BðjÞð Þ ¼ :TRUE:. If either one of a complex conjugate pair is
selected, then both complex generalized eigenvalues are selected.

Note that in the ill-conditioned case, a selected complex generalized eigenvalue may no
longer satisfy SELCTG ARðjÞ;AIðjÞ;BðjÞð Þ ¼ :TRUE: after ordering. INFO ¼ Nþ 2 in
this case.

SELCTG must either be a module subprogram USEd by, or declared as EXTERNAL in, the
(sub)program from which F08XAF (DGGES) is called. Parameters denoted as Input must not be
changed by this procedure.

5: N – INTEGER Input

On entry: n, the order of the matrices A and B.

Constraint: N � 0.

6: AðLDA,�Þ – REAL (KIND=nag_wp) array Input/Output

Note: the second dimension of the array A must be at least max 1;Nð Þ.
On entry: the first of the pair of matrices, A.

On exit: A has been overwritten by its generalized Schur form S.

7: LDA – INTEGER Input

On entry: the first dimension of the array A as declared in the (sub)program from which F08XAF
(DGGES) is called.

Constraint: LDA � max 1;Nð Þ.

8: BðLDB,�Þ – REAL (KIND=nag_wp) array Input/Output

Note: the second dimension of the array B must be at least max 1;Nð Þ.
On entry: the second of the pair of matrices, B.

On exit: B has been overwritten by its generalized Schur form T .

9: LDB – INTEGER Input

On entry: the first dimension of the array B as declared in the (sub)program from which F08XAF
(DGGES) is called.

Constraint: LDB � max 1;Nð Þ.

10: SDIM – INTEGER Output

On exit: if SORT ¼ N , SDIM ¼ 0.

If SORT ¼ S , SDIM ¼ number of eigenvalues (after sorting) for which SELCTG is .TRUE..
(Complex conjugate pairs for which SELCTG is .TRUE. for either eigenvalue count as 2.)

11: ALPHARðNÞ – REAL (KIND=nag_wp) array Output

On exit: see the description of BETA.

12: ALPHAIðNÞ – REAL (KIND=nag_wp) array Output

On exit: see the description of BETA.

13: BETAðNÞ – REAL (KIND=nag_wp) array Output

On exit: ALPHARðjÞ þ ALPHAIðjÞ � ið Þ=BETAðjÞ, for j ¼ 1; 2; . . . ;N, will be the generalized
eigenvalues. ALPHARðjÞ þ ALPHAIðjÞ � i, and BETAðjÞ, for j ¼ 1; 2; . . . ;N, are the diagonals
of the complex Schur form S; Tð Þ that would result if the 2 by 2 diagonal blocks of the real Schur
form of A;Bð Þ were further reduced to triangular form using 2 by 2 complex unitary
transformations.

If ALPHAIðjÞ is zero, then the jth eigenvalue is real; if positive, then the jth and jþ 1ð Þst
eigenvalues are a complex conjugate pair, with ALPHAIðjþ 1Þ negative.

Note: the quotients ALPHARðjÞ=BETAðjÞ and ALPHAIðjÞ=BETAðjÞ may easily overflow or
underflow, and BETAðjÞ may even be zero. Thus, you should avoid naively computing the ratio

�=�. However, ALPHAR and ALPHAI will always be less than and usually comparable with Ak k2
in magnitude, and BETA will always be less than and usually comparable with Bk k2.

14: VSLðLDVSL,�Þ – REAL (KIND=nag_wp) array Output

Note: the second dimension of the array VSL must be at least max 1;Nð Þ if JOBVSL ¼ V , and at
least 1 otherwise.

On exit: if JOBVSL ¼ V , VSL will contain the left Schur vectors, Q.

If JOBVSL ¼ N , VSL is not referenced.

15: LDVSL – INTEGER Input

On entry: the first dimension of the array VSL as declared in the (sub)program from which F08XAF
(DGGES) is called.

Constraints:

if JOBVSL ¼ V , LDVSL � max 1;Nð Þ;
otherwise LDVSL � 1.

16: VSRðLDVSR,�Þ – REAL (KIND=nag_wp) array Output

Note: the second dimension of the array VSR must be at least max 1;Nð Þ if JOBVSR ¼ V , and at
least 1 otherwise.

On exit: if JOBVSR ¼ V , VSR will contain the right Schur vectors, Z.

If JOBVSR ¼ N , VSR is not referenced.

17: LDVSR – INTEGER Input

On entry: the first dimension of the array VSR as declared in the (sub)program from which F08XAF
(DGGES) is called.

Constraints:

if JOBVSR ¼ V , LDVSR � max 1;Nð Þ;
otherwise LDVSR � 1.

18: WORKðmax 1;LWORKð ÞÞ – REAL (KIND=nag_wp) array Workspace

On exit: if INFO ¼ 0, WORKð1Þ contains the minimum value of LWORK required for optimal
performance.

19: LWORK – INTEGER Input

On entry: the dimension of the array WORK as declared in the (sub)program from which F08XAF
(DGGES) is called.

If LWORK ¼ �1, a workspace query is assumed; the routine only calculates the optimal size of the
WORK array, returns this value as the first entry of the WORK array, and no error message related
to LWORK is issued.

Suggested value: for optimal performance, LWORK must generally be larger than the mimimum;
add, say nb� N, where nb is the optimal block size.

Constraints:

if N ¼ 0, LWORK � 1;
otherwise LWORK � max 8� N; 6� Nþ 16ð Þ.

20: BWORKð�Þ – LOGICAL array Workspace

Note: the dimension of the array BWORK must be at least 1 if SORT ¼ N , and at least max 1;Nð Þ
otherwise.

If SORT ¼ N , BWORK is not referenced.

21: INFO – INTEGER Output

On exit: INFO ¼ 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO ¼ �i, argument i had an illegal value. An explanatory message is output, and execution of
the program is terminated.

INFO ¼ 1 to N

The QZ iteration failed. A;Bð Þ are not in Schur form, but ALPHARðjÞ, ALPHAIðjÞ, and
BETAðjÞ should be correct for j ¼ INFOþ 1; . . . ;N.

INFO ¼ Nþ 1

Unexpected error returned from F08XEF (DHGEQZ).

INFO ¼ Nþ 2

After reordering, roundoff changed values of some complex eigenvalues so that leading eigenvalues
in the generalized Schur form no longer satisfy SELCTG ¼ :TRUE:. This could also be caused by
underflow due to scaling.

INFO ¼ Nþ 3

The eigenvalues could not be reordered because some eigenvalues were too close to separate (the
problem is very ill-conditioned).

7 Accuracy

The computed generalized Schur factorization satisfies

Aþ E ¼ QSZT, Bþ F ¼ QTZT,

where

E;Fð Þk kF ¼ O �ð Þ A;Bð Þk kF
and � is the machine precision. See Section 4.11 of Anderson et al. (1999) for further details.

8 Further Comments

The total number of floating point operations is proportional to n3.

The complex analogue of this routine is F08XNF (ZGGES).

9 Example

This example finds the generalized Schur factorization of the matrix pair A;Bð Þ, where

A ¼

3:9 12:5 �34:5 �0:5
4:3 21:5 �47:5 7:5
4:3 21:5 �43:5 3:5
4:4 26:0 �46:0 6:0

0
BB@

1
CCA and B ¼

1:0 2:0 �3:0 1:0
1:0 3:0 �5:0 4:0
1:0 3:0 �4:0 3:0
1:0 3:0 �4:0 4:0

0
BB@

1
CCA,

such that the real eigenvalues of A;Bð Þ correspond to the top left diagonal elements of the generalized
Schur form, S; Tð Þ.

Note that the block size (NB) of 64 assumed in this example is not realistic for such a small problem, but
should be suitable for large problems.

9.1 Program Text

! F08XAF Example Program Text
! Mark 24 Release. NAG Copyright 2012.

Module f08xafe_mod

! F08XAF Example Program Module:
! Parameters and User-defined Routines

! .. Use Statements ..
Use nag_library, Only: nag_wp

! .. Implicit None Statement ..
Implicit None

! .. Parameters ..
Integer, Parameter :: nb = 64, nin = 5, nout = 6

Contains
Function selctg(ar,ai,b)

! Logical function selctg for use with DGGES (F08XAF)

! Returns the value .TRUE. if the eigenvalue is real

! .. Function Return Value ..
Logical :: selctg

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: ai, ar, b

! .. Local Scalars ..
Logical :: d

! .. Executable Statements ..
If (ai==0.0E0_nag_wp) Then

d = .True.
Else

d = .False.
End If

selctg = d

Return
End Function selctg

End Module f08xafe_mod
Program f08xafe

! F08XAF Example Main Program

! .. Use Statements ..
Use nag_library, Only: dgemm, dgges, dlange => f06raf, nag_wp, x02ajf, &

x04caf
Use f08xafe_mod, Only: nb, nin, nout, selctg

! .. Implicit None Statement ..
Implicit None

! .. Local Scalars ..
Real (Kind=nag_wp) :: alph, bet, normd, norme
Integer :: i, ifail, info, lda, ldb, ldc, &

ldd, lde, ldvsl, ldvsr, lwork, &
n, sdim

! .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: a(:,:), alphai(:), alphar(:), &

b(:,:), beta(:), c(:,:), d(:,:), &
e(:,:), vsl(:,:), vsr(:,:), &
work(:)

Real (Kind=nag_wp) :: dummy(1)
Logical, Allocatable :: bwork(:)

! .. Intrinsic Procedures ..
Intrinsic :: max, nint

! .. Executable Statements ..
Write (nout,*) ’F08XAF Example Program Results’

Write (nout,*)
Flush (nout)

! Skip heading in data file
Read (nin,*)
Read (nin,*) n
lda = n
ldb = n
ldc = n
ldd = n
lde = n
ldvsl = n
ldvsr = n
Allocate (a(lda,n),alphai(n),alphar(n),b(ldb,n),beta(n),vsl(ldvsl,n), &

vsr(ldvsr,n),bwork(n),c(ldc,n),d(ldd,n),e(lde,n))

! Use routine workspace query to get optimal workspace.
lwork = -1

! The NAG name equivalent of dgges is f08xaf
Call dgges(’Vectors (left)’,’Vectors (right)’,’Sort’,selctg,n,a,lda,b, &

ldb,sdim,alphar,alphai,beta,vsl,ldvsl,vsr,ldvsr,dummy,lwork,bwork, &
info)

! Make sure that there is enough workspace for blocksize nb.
lwork = max(8*n+16+n*nb,nint(dummy(1)))
Allocate (work(lwork))

! Read in the matrices A and B
Read (nin,*)(a(i,1:n),i=1,n)
Read (nin,*)(b(i,1:n),i=1,n)

! Copy A and B into D and E respectively
d(1:n,1:n) = a(1:n,1:n)
e(1:n,1:n) = b(1:n,1:n)

! Print Matrix A and Matrix B
! ifail: behaviour on error exit
! =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft

ifail = 0
Call x04caf(’General’,’ ’,n,n,a,lda,’Matrix A’,ifail)
Write (nout,*)
Call x04caf(’General’,’ ’,n,n,b,ldb,’Matrix B’,ifail)
Write (nout,*)

! Find the generalized Schur form
! The NAG name equivalent of dgges is f08xaf

Call dgges(’Vectors (left)’,’Vectors (right)’,’Sort’,selctg,n,a,lda,b, &
ldb,sdim,alphar,alphai,beta,vsl,ldvsl,vsr,ldvsr,work,lwork,bwork,info)

If (info>0 .And. info/=(n+2)) Then
Write (nout,99999) ’Failure in DGGES. INFO =’, info

Else
! Compute A - Q * S * Z^T from Schur factorization of (A,B) and store in
! matrix D
! The NAG name equivelent of dgemm is f06yaf

alph = 1.0_nag_wp
bet = 0.0_nag_wp
Call dgemm(’N’,’N’,n,n,n,alph,vsl,ldvsl,a,lda,bet,c,ldc)
alph = -1.0_nag_wp
bet = 1.0_nag_wp
Call dgemm(’N’,’T’,n,n,n,alph,c,ldc,vsr,ldvsr,bet,d,ldd)

! Compute B - Q * T * Z^T from Schur factorization of (A,B) and store in
! matrix E

alph = 1.0_nag_wp
bet = 0.0_nag_wp
Call dgemm(’N’,’N’,n,n,n,alph,vsl,ldvsl,b,ldb,bet,c,ldc)
alph = -1.0_nag_wp
bet = 1.0_nag_wp
Call dgemm(’N’,’T’,n,n,n,alph,c,ldc,vsr,ldvsr,bet,e,lde)

! Find norms of matrices D and E and warn if either is too large
! f06raf is the NAG name equivalent of the LAPACK auxiliary dlange

normd = dlange(’O’,ldd,n,d,ldd,work)
norme = dlange(’O’,lde,n,e,lde,work)
If (normd>x02ajf()**0.8_nag_wp .Or. norme>x02ajf()**0.8_nag_wp) Then

Write (nout,*) ’Norm of A-(Q*S*Z^T) or norm of B-(Q*T*Z^T) &
&is much greater than 0.’

Write (nout,*) ’Schur factorization has failed.’
Else

Write (nout,99999) &
’Number of eigenvalues for which SELCTG is true = ’, sdim

Write (nout,*)

! Print the generalized eigenvalues
Write (nout,*) ’Selected generalized eigenvalues’

Do i = 1, sdim
If (beta(i)/=0.0_nag_wp) Then

Write (nout,99998) i, ’(’, alphar(i)/beta(i), ’,’, &
alphai(i)/beta(i), ’)’

Else
Write (nout,99997) i

End If
End Do
If (info==(n+2)) Then

Write (nout,99996) ’***Note that rounding errors mean ’, &
’that leading eigenvalues in the generalized’, &
’Schur form no longer satisfy SELCTG = .TRUE.’

Write (nout,*)
End If

End If
Flush (nout)

End If

99999 Format (1X,A,I4)
99998 Format (1X,I4,5X,A,F7.3,A,F7.3,A)
99997 Format (1X,I4,’Eigenvalue is infinite’)
99996 Format (1X,2A/1X,A)

End Program f08xafe

9.2 Program Data

F08XAF Example Program Data
4 :Value of N
3.9 12.5 -34.5 -0.5
4.3 21.5 -47.5 7.5
4.3 21.5 -43.5 3.5
4.4 26.0 -46.0 6.0 :End of matrix A
1.0 2.0 -3.0 1.0
1.0 3.0 -5.0 4.0
1.0 3.0 -4.0 3.0
1.0 3.0 -4.0 4.0 :End of matrix B

9.3 Program Results

F08XAF Example Program Results

Matrix A
1 2 3 4

1 3.9000 12.5000 -34.5000 -0.5000
2 4.3000 21.5000 -47.5000 7.5000
3 4.3000 21.5000 -43.5000 3.5000
4 4.4000 26.0000 -46.0000 6.0000

Matrix B
1 2 3 4

1 1.0000 2.0000 -3.0000 1.0000
2 1.0000 3.0000 -5.0000 4.0000
3 1.0000 3.0000 -4.0000 3.0000
4 1.0000 3.0000 -4.0000 4.0000

Number of eigenvalues for which SELCTG is true = 2

Selected generalized eigenvalues
1 ( 2.000, 0.000)
2 ( 4.000, 0.000)

___________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________

